

Скриптовый язык контроллера CLRC-02

Скрипт - это текстовый файл, содержащий строки с условиями и командами. Исполнение скрипта состоит в последовательной интерпретации строк, проверке условий и выполнении команд. Если обработаны все строки, то скрипт считается исполненным. Пустые строки игнорируются. Игнорируются пробелы и знаки табуляции. Всё, что в строке расположено правее знаков `//` считается комментарием и игнорируется.

При запуске скрипта выполняется его проверка на корректность команд. В случае выявления ошибки, скрипт не будет запущен и на LCD-дисплее отобразится номер строки с ошибкой, содержание строки и расшифровка ошибки.

По окончании исполнения скрипта мощность нагревателя устанавливается в 0 и закрываются все клапаны.

Далее следует описание условий и команд скриптового языка.

h=<value> - задать заголовок. **<value>** - строка от одного до пяти знаков. Пробелы игнорируются. Заголовок отображается на LCD-дисплее во время исполнения скрипта и нужен для обозначения текущего состояния процесса.

Пример:

```
//задать заголовок  
h = R-TH
```

t<index>s=<value> - привязать к логическому термометру с номером **<index>** физический термометр с номером **<value>**. **<index>** может принимать целочисленные значения от 1 до 3. **<value>** может принимать целочисленные значения от 1 до 4. Номера физических термометров можно узнать выбрав в главном меню пункт **Check sensors**. Если термометр привязан, то в процессе исполнения скрипта его значение будет отображаться на LCD-дисплее.

Пример:

```
// привязать к логическому термометру t1 второй физический  
t1s = 2
```

t<index>c=<value> - корректировать значения термометра с номером <index> на значение <value>. <index> может принимать целочисленные значения от 1 до 3. <value> - вещественное число в диапазоне от -10 до +10. До этой команды должен быть задан корректируемый термометр (см. t<index>s=<value>).

Пример:

```
//корректировать термометр t1 на -0.7 градуса
t1c = -0.7
```

dt=<exp> - создать логический термометр разности температур. <exp> может принимать следующие значения: t1-t2, t1-t3, t2-t1, t2-t3, t3-t1, t3-t2. До этой команды должны быть заданы используемые в <exp> термометры (см. t<index>s=<value>). Если термометр dt создан, то в процессе исполнения скрипта его значение будет отображаться на LCD-дисплее.

Пример:

```
//создать термометр разности температур (t2 - t1)
dt = t2 - t1
```

r<index>=<value> - установить состояние реле с номером <index>. <index> должен быть целочисленным значением в диапазоне от 1 до 8. <value> может принимать значение 0 (реле выключено) и 1 (реле включено). В процессе исполнения скрипта на LCD-дисплее отображается состояние всех имеющихся клапанов (знак "x" - клапан закрыт, знак "." - клапан открыт).

Пример:

```
//включить третье реле
r3 = 1
```

p=<value> - установить мощность нагревателя. <value> может принимать целочисленные значения от 0 до 100. В процессе исполнения скрипта на LCD-дисплее отображается текущее значение мощности.

Пример:

```
//установить мощность в 57% от максимальной
p = 57
```

p(t<index>)=linear(<minT>, <maxT>, <minP>, <maxP>) - задать линейный режим изменения мощности в зависимости от значений термометра с номером <index>. При температуре <minT> мощность будет равна <minP>. При температуре <maxT> мощность будет равна <maxP>. При прочих значениях термометра мощность будет вычисляться пропорционально указанным значениям диапазонов температуры и мощности. Значения всех параметров должны быть в интервале от 0 до 100. Значение <minT> должно быть меньше значения <maxT>. Значение <minP> должно быть меньше значения <maxP>.

Пример:

```
//изменять мощность пропорционально значению t1
//при 80 градусах мощность должна быть 60%
//при увеличении значения t1 мощность увеличивается
//при 98 градусах мощность будет 70%
p(t1) = linear(80, 98, 60, 70)
```

wait - ожидать прерывания скрипта пользователем. Выполнение команды останавливает общий таймер скрипта.

Пример:

```
//ожидать прерывания от пользователя
Wait
```

d=<value> - пауза в выполнении скрипта в <value> минут. <value> - целочисленное значение в диапазоне от 1 до 60.

Пример:

```
//пауза 20 минут
d = 20
```

<exp><value>[:<durability>] - проверка значения <exp>. Условие считается выполненным, если значение <exp> больше значения <value>. Если задан необязательный параметр <durability>, то условие считается выполненным, если значение <exp> больше значения <value> в течении <durability> минут. <exp> может принимать следующие значения: t_1 , t_2 , t_3 , dt , $(t_1 - t_2)$, $(t_1 - t_3)$, $(t_2 - t_1)$, $(t_2 - t_3)$, $(t_3 - t_1)$, $(t_3 - t_2)$. Значение <value> должно быть больше 0 и меньше 100. Значение <durability> должно быть целочисленным в диапазоне от 1 до 10.

Примеры:

```
//ждать, пока температура t3 не превысит 80 градусов
t3 > 80

//ждать, пока разность (t2 - t1)
// не превысит 0.5 градуса
(t2 - t1) > 0.5

//ждать, пока заданная ранее разность термометров
// не превысит 1.2 градуса
dt > 1.2

//ждать, пока температура t2 не будет превышать 60
// градусов в течении 5 минут
t2 > 60 : 5
```

<exp><value>[:<durability>] - проверка значения <exp>. Аналогично предыдущему условию, но значение <exp> должно быть меньше значения <value>.

Примеры:

```
//ждать, пока температура t3 не опустится
// ниже 50 градусов
t3 < 50

//ждать, пока разность (t3 - t1)
// не будет меньше 0.8 градуса
(t3 - t1) < 0.8

//ждать, пока заданная ранее разность термометров
// не будет меньше 0.4 градуса
dt < 0.4

//ждать, пока температура t1 не будет меньше
// 40 градусов в течении 3 минут
t1 < 40 : 3
```

Примеры скриптов

Скрипт дистилляции:

```
//термометр t1 в дефлегматоре
t1s = 1

//мощность нагревателя на 100%
p = 100

//ждать выдоха
t1 > 70

//открыть клапан подачи воды в дефлегматор
r1 = 1

//открыть клапан отбора
r2 = 1

//ждать, когда в дефлегматоре температура
// достигнет 98 градусов
t1 > 98

//перекрыть отбор
r2 = 0

//выключить нагреватель
p = 0

//подождать 20 минут
d = 20

//перекрыть подачу воды в дефлегматор
r1 = 0

//ждать прерывания от пользователя
wait
```

Скрипт ректификации:

```
//термометр t1 в дефлегматоре
t1s = 1

//термометр врёт на 0.7 градуса. коррекция
t1c = -0.7

//термометр t2 на царге в 40 см от низа дефлегматора
t2s = 3

//термометр врёт на -0.2 градуса. коррекция
t2c = 0.2

//термометр t3 в баке
t3s = 2

//мощность нагревателя на 100%
p = 100

//ждать, когда бак прогреется до 75 градусов
t3 > 75

//открыть клапан подачи воды в дефлегматор
r1 = 1

//мощность на 50%
p = 50

//подождать 10 минут для стабилизации колонны
d = 10

//настроить автоматическое увеличение мощности
// нагревателя от 50% до 70% по мере выпаривания спирта
p(t3) = linear(80, 98, 50, 70)

//подождать еще 10 минут для стабилизации колонны
d = 10

//нужно следить за разницей (t2 - t1)
dt = t2 - t1

//если dt < 0.3 в течении 5 минут,
// то можно начинать отбор
dt < 0.3 : 5

//открыть клапан отбора голов
r2 = 1

//отбирать головы до dt > 0.5 в течении 2 минут
dt > 0.5 : 2

//закрыть клапан отбора голов
```

```
r2 = 0

//открыть клапан отбора тела
// (у него другая скорость отбора)
r3 = 1

//отбирать до прихода хвостов
dt > 1

//перекрыть отбор тела
r3 = 0

//выключить нагреватель
p = 0

//подождать 20 минут
d = 20

//перекрыть подачу воды в дефлегматор
r1 = 0

//ждать прерывания от пользователя
wait
```